// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/extensions/IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.3) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    /**
     * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    /**
     * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
     * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
     */
    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    /**
     * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 oldAllowance = token.allowance(address(this), spender);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance + value));
    }

    /**
     * @dev Decrease the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance - value));
        }
    }

    /**
     * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful. Meant to be used with tokens that require the approval
     * to be set to zero before setting it to a non-zero value, such as USDT.
     */
    function forceApprove(IERC20 token, address spender, uint256 value) internal {
        bytes memory approvalCall = abi.encodeWithSelector(token.approve.selector, spender, value);

        if (!_callOptionalReturnBool(token, approvalCall)) {
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, 0));
            _callOptionalReturn(token, approvalCall);
        }
    }

    /**
     * @dev Use a ERC-2612 signature to set the `owner` approval toward `spender` on `token`.
     * Revert on invalid signature.
     */
    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        require(returndata.length == 0 || abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     *
     * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
     */
    function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
        // and not revert is the subcall reverts.

        (bool success, bytes memory returndata) = address(token).call(data);
        return
            success && (returndata.length == 0 || abi.decode(returndata, (bool))) && Address.isContract(address(token));
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/interfaces/NFAiGatewayInterface.sol



// SPDX-License-Identifier: MIT
pragma solidity =0.8.21;

/** 
* @title NFAiGatewayInterface
* @notice Interface for the NFAiStakingLottery contract
* @dev Defines the methods and custom errors for the NFAiStakingLottery contract
*/
interface NFAiGatewayInterface {

    /**
    * @notice Custom Events
    */
    event Staked(address indexed staker, uint256 amount);
    event Unstaked(address indexed staker, uint256 amount);
    event WinnerDrawn(address indexed winner, uint256 amount);
    event RandomnessRequested(uint256 requestId);
    event RandomnessFulfilled(uint256 requestId, uint256 randomResult);
    event ParticipationConfirmed(address indexed user);
    event PrizeTokensDeposited(address indexed owner, uint256 amount);

    /**
    * @notice Custom Errors
    */
    error AbnormalRandomResult();
    error InsufficientEligibleParticipants();
    error TotalWeightInsuffiency();
    error ZeroAdrWinner();
    error ZeroAmount();
    error RequestRandomnessFailed();
    error NoRandomnessProvided();
    error UnmatchedRequest();
    error NotUser(address caller);
    error TransferFailed();

    /**
    * @notice Wrapper function for staking tokens
    * @param amount Amount of tokens to stake
    * @param user Address of the user
    */
    function stake(uint256 amount, address user) external;

    /**
    * @notice Wrapper function for unstaking tokens
    * @param user Address of the user
    */
    function unstake(address user) external;

    /**
    * @notice Wrapper function for drawing winners
    * @param _drawingStrategy The drawing strategy to use
    * @param _numberOfWinners The number of winners to draw
    */
    function draw(uint8 _drawingStrategy, uint256 _numberOfWinners) external;

    /** 
    * @notice Wrapper function for refreshing the staking time.
    * @param user Address of the user
    * @dev This function is used to refresh the staking time for the caller.
    */ 
    function confirmParticipation(address user) external;

    /**
    * @dev Set the minimum staking time.
    * @param _minutes Minimum staking time in minutes.
    */
    function setMinimumStakeTime(uint256 _minutes) external;

    /**
    * @param ethTO Address to send the ETH to.
    * @param ercTO Address to send the ERC to.
    * @param tokenAdr Address of the token to withdraw.
    * @dev Emergency function to withdraw stuck ETH/ERC.
    */
    function emergencyWithdraw(address ethTO, address ercTO, address tokenAdr) external;

    /**
    * @notice Get the stakers addresses
    * @return Array of addresses of all stakers
    */
    function getStakers() external view returns (address[] memory);

    /** 
    * @notice Get the staked amount for a specific user
    * @param user Address of the user
    * @return Amount of tokens staked by the user
    */ 
    function getStakerAmount(address user) external view returns (uint256);

    /**
    * @notice Get the staked time for a specific user
    * @param user Address of the user
    * @return Timestamp of the last stake
    */
    function getStakerTime(address user) external view returns (uint256);

    /** 
    * @notice Get the total staked amount
    * @return Total amount of tokens staked
    */
    function getTotalStaked() external view returns (uint256);

    /**
    * @notice Get the total eligible staked amount
    * @return Total amount of tokens staked by eligible users
    */
    function getTotalEligibleStaked() external view returns (uint256);

    /** 
    * @notice Get the last winner
    * @return Address of the last winner
    * @return Amount won by the last winner
    * @return Timestamp of the last winner
    */
    function getLastWinnerInfo() external view returns (address, uint256, uint256);

    /**
    * @notice Get the last winner amount
    * @return All addresses of all historical winners
    * @return All amounts won by all historical winners
    */
    function getAllWinnerInfo() external view returns (address[] memory, uint256[] memory);
    /**
    * @notice Get the minimum stake time to be eligible for the lottery
    * @return Minimum stake time in minutes
    */
    function getMinimumStakeTime() external view returns (uint256);

    /**
    * @notice Get the eligibility of all stakers
    * @return Returns a single entity back.
    */
    function getEligibility(address user) external view returns (bool);

    /**
    * @notice Get the odds of winning for a specific user
    * @param user Address of the user
    * @return Odds of winning for the user as a percentage multiplied by 10^4
    */
    function getUserOdds(address user) external view  returns (uint256);

    /**
    * @notice Get ERC20 Token Address Registered for Staking
    * @return Address registered for IERC20
    */
    function getRegisteredToken() external view returns (address);

    /**
    * @notice Set the initial values for the contract
    * @param _keyHash The key hash for the Chainlink VRF Coordinator
    * @param subscriptionId The subscription ID for the Chainlink VRF Coordinator
    * @param _callbackGasLimit The callback gas limit for the Chainlink VRF Coordinator
    * @param _minimumRequestConfirmations The minimum request confirmations for the Chainlink VRF Coordinator
    * @param _numWords The number of words for the Chainlink VRF Coordinator
    * @param _token Address of the token to use for staking
    */
    function setInitializingFactors(
        bytes32 _keyHash, 
        uint64 subscriptionId, 
        uint32 _callbackGasLimit, 
        uint16 _minimumRequestConfirmations, 
        uint32 _numWords, 
        address _token
    ) external;
}
// File: contracts/IPWrapper.sol



// SPDX-License-Identifier: MIT
pragma solidity =0.8.21;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "./interfaces/NFAiGatewayInterface.sol"; 

/**
* @title IPWrapper
* @dev This contract is a wrapper for the NFAiStakingLottery contract. It is used to make the NFAiStakingLottery contract compatible with the IP contract. 
*/
contract IPWrapper is NFAiGatewayInterface, Ownable {
    using SafeERC20 for IERC20;
    NFAiGatewayInterface public nFAiStakingLottery;
    
    constructor() {
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getRegisteredToken() external view override returns (address) {
        return nFAiStakingLottery.getRegisteredToken();
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getAllWinnerInfo() external view override returns (address[] memory, uint256[] memory) {
        return nFAiStakingLottery.getAllWinnerInfo();
    }
    
    /** 
    * @inheritdoc NFAiGatewayInterface
    */
    function getLastWinnerInfo() external view override returns (address, uint256, uint256) {
        return nFAiStakingLottery.getLastWinnerInfo();
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getStakerAmount(address user) external view override returns (uint256) {
        return nFAiStakingLottery.getStakerAmount(user);
    }                                  

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getMinimumStakeTime() external view override returns (uint256) {
        return nFAiStakingLottery.getMinimumStakeTime();
    }

    function getEligibility(address user) external view override returns (bool) {
        return nFAiStakingLottery.getEligibility(user);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getTotalEligibleStaked() external  view returns (uint256) {
        return nFAiStakingLottery.getTotalEligibleStaked();
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getStakers() external view returns (address[] memory) {
        return nFAiStakingLottery.getStakers();
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getTotalStaked() external view  returns (uint256){
        return nFAiStakingLottery.getTotalStaked();
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getUserOdds(address user) external view override returns (uint256) {
        return nFAiStakingLottery.getUserOdds(user);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function getStakerTime(address user) external view override returns (uint256) {
        return nFAiStakingLottery.getStakerTime(user);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function stake(uint256 amount, address user) external override {
        if(msg.sender != user) {
            revert NotUser(msg.sender);
        }
        IERC20 token = IERC20(nFAiStakingLottery.getRegisteredToken());
        token.safeTransferFrom(msg.sender, address(this), amount);
        token.safeIncreaseAllowance(address(nFAiStakingLottery), amount);
        nFAiStakingLottery.stake(amount, user);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function unstake(address user) external override {
        if(msg.sender != user) {
            revert NotUser(msg.sender);
        }
        nFAiStakingLottery.unstake(user);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function confirmParticipation(address user) external override {
        if(msg.sender != user) {
            revert NotUser(msg.sender);
        }
        nFAiStakingLottery.confirmParticipation(user);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function setMinimumStakeTime(uint256 _minutes) external override onlyOwner {
        nFAiStakingLottery.setMinimumStakeTime(_minutes);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function emergencyWithdraw(address ethTO, address ercTO, address tokenAdr) external override onlyOwner {
        nFAiStakingLottery.emergencyWithdraw(ethTO, ercTO, tokenAdr);
    }

    /**
    * @inheritdoc NFAiGatewayInterface
    */
    function draw(uint8 _drawingStrategy, uint256 _numberOfWinners) external override onlyOwner {
        nFAiStakingLottery.draw(_drawingStrategy, _numberOfWinners);
    }

    /**
    * @notice Set the NFAiStakingLotteryAddress
    * @param _nFAiStakingLottery Address of the NFAiStakingLottery contract
    */
    function setNFAiStakingLottery(address _nFAiStakingLottery) external onlyOwner {
        nFAiStakingLottery = NFAiGatewayInterface(_nFAiStakingLottery);
    }

    /** 
    * @inheritdoc NFAiGatewayInterface
    */
    function setInitializingFactors(
        bytes32 _keyHash, 
        uint64 subscriptionId, 
        uint32 _callbackGasLimit, 
        uint16 _minimumRequestConfirmations, 
        uint32 _numWords, 
        address _token
    ) external override onlyOwner {
        nFAiStakingLottery.setInitializingFactors(
            _keyHash, 
            subscriptionId, 
            _callbackGasLimit, 
            _minimumRequestConfirmations, 
            _numWords, 
            _token
        );
    }
}